

Simulación de fallas en sistemas de potencia usando componentes simétricos y el programa PSPICE

Horacio A. García Correa

Vicente Rivera Velázquez

Candidatos a graduación en ingeniería eléctrica, UPR

Sinopsis

En este artículo se discute cómo usar el programa PSPICE para analizar fallas en sistemas de potencia por el método de componentes simétricos. Sólo se discuten las reglas, instrucciones y restricciones de PSPICE necesarias para simular fallas simétricas y asimétricas. Se presentan ilustraciones para ayudar al lector a entender cómo se interconectan los componentes de secuencia positiva, negativa y cero de una falla doble de línea a tierra.

Fault simulation in power systems using symmetric components and the computer program PSPICE

Abstract

This article shows how to use the computer program PSPICE to analyze faults in power systems by the symmetrical components method. The paper covers only the rules, instructions and limitations of PSPICE necessary for both symmetrical and unsymmetrical fault simulations. Figures help the reader to understand how the positive, negative and zero components sequence are interconnected in a double line to ground fault.

Introducción

SPICE es un programa de simulación que le permite al ingeniero diseñar y simular circuitos eléctricos en una computadora. SPICE es una abreviatura en inglés de "Simulation Program with Integrated Circuit Emphasis" desarrollado en los laboratorios de investigación electrónica de la Universidad de California, Berkeley, en la década de los 70.

García y Rivera/Simulación fallas componentes simétricos

La versión de PSPICE es la primera versión que se introduce en el mercado para computadoras personales. Aunque su algoritmo interno difiere del algoritmo del SPICE original, su funcionamiento es bastante similar al SPICE de Berkeley. PSPICE tiene algunas instrucciones únicas que no son compatibles con SPICE, pero estas instrucciones son para casos particulares. Después que el usuario las identifique y las evite puede trabajar con PSPICE exactamente como con la versión original de SPICE.

Pasos a seguir para simular fallas en sistemas de potencia con PSPICE

Dibujar el diagrama monolineal del sistema de potencia.

La representación del circuito a simularse se compone de una serie de líneas que describen la topología del circuito, la interconexión de los elementos del circuito entre nodos enumerados y los valores asignados a los parámetros (resistencia, inductancia, capacitancia y otros) de cada elemento. La figura 1 presenta el diagrama monolineal del sistema de potencia a utilizar como ejemplo para simular fallas. Este diagrama contiene máquinas de rotación que pueden aportar corriente a la falla, transformadores con el tipo de conexión de los embobinados primarios y secundarios, líneas de transmisión y cualquier otro elemento necesario para el análisis.

Por lo general, las impedancias se expresan en porcentaje o en por unidad. Para realizar el análisis de corto circuito es necesario expresar cada impedancia en por unidad de una misma base. Los valores base para este sistema son 13.8 kV_{L-L} y $100 \text{ MVA}_{3\phi}$.

Dibujar los circuitos de secuencia positiva, negativa y cero del sistema de potencia (figuras 2, 3 y 4, respectivamente)

Se usa el número o un símbolo para representar el valor de los elementos. La tabla 1 resume los factores simbólicos y su forma exponencial correspondiente.




Figura 1. Diagrama monolineal del circuito a analizar

García y Rivera/Simulación fallas componentes simétricos

Tabla 1. Factores simbólicos y forma exponencial

Símbolo	Forma exponencial	Valor
F	1 E-15	10^{-15}
P	1 E-12	10^{-12}
N	1 E-9	10^{-9}
U	1 E-6	10^{-6}
M	1 E-3	10^{-3}
K	1 E+3	10^3
MEG	1 E+6	10^6
G	1 E+9	10^9
T	1 E+12	10^{12}

Para identificar cada elemento se usa el siguiente formato: resistencia (Rxxx), inductor (Lxxx), condensador (Cxxx), fuente de voltaje (Vxxx) y una fuente de corriente (Ixxx).

Modificar los circuitos

Añadir fuentes de voltaje CC de cero voltios en serie con los elementos en que se requiera medir los componentes simétricos de la corriente.

Una fuente de voltaje CC de cero voltios no altera el circuito porque equivale a un corto circuito o un alambre. Las figuras 5, 6 y 7 presentan los circuitos de secuencia positiva, negativa y cero modificados para analizarlos con el programa PSPICE. Observe que se añaden tres fuentes de voltaje CC de cero voltios en cada circuito de secuencia para medir los componentes simétricos de las corrientes que fluyen en tres puntos diferentes del sistema de potencia.




Figura 2. Circuito monofásico de secuencia positiva.




Figura 3. Circuito monofásico de secuencia negativa.




Figura 4. Circuito de secuencia cero.

García y Rivera/Simulación fallas componentes simétricos

Por ejemplo, la fuente de voltaje VD1 aparece en los tres circuitos de secuencia para medir el componente positivo, negativo y cero de la corriente desbalanceada que fluye desde el nodo 1 al 6 del sistema de potencia de la figura 1. De igual forma, VD2 y VD3 miden los componentes simétricos de la corriente que fluye del nodo 4 al 6 y del nodo 5 al 6 del sistema, respectivamente.

Los nombres dados a estas tres fuentes de voltaje "CC" se repiten en los circuitos de secuencia. Posteriormente se crea un archivo que define los circuitos de secuencia positiva, negativa y cero como subcircuitos. Un subcircuito permite que los nombres o números asignados a cualquier elemento o nodo se puedan repetir en otro subcircuito sin crear conflictos entre ambos. El programa PSPICE distingue el nombre dado a la fuente de voltaje en los tres circuitos de secuencia durante el análisis y le añade al nombre de la fuente la instrucción que llama el subcircuito Xn. Por ejemplo: X1.VD1, X2.VD1 y X3.VD1 donde X1, X2 y X3 determinan el subcircuito que contiene la fuente de voltaje a la cual el programa PSPICE hace referencia durante el análisis. En este caso X1 se utiliza para llamar el subcircuito que describe la topología de la secuencia positiva, X2 la topología de secuencia negativa y X3 la topología de secuencia cero.

Enumerar todos los nodos en los circuitos de secuencia

PSPICE se basa en un análisis de nodos. Los nodos deben enumerarse con números positivos, pero éstos no tienen que estar en secuencia.

Los circuitos de secuencia positiva, negativa y cero se definen más adelante como subcircuitos. El nodo de referencia en cada circuito de secuencia se selecciona como nodo externo en la instrucción que define los subcircuitos (.SUBCKT). La única restricción para seleccionar los nodos externos es que el número cero no se puede usar. Por lo tanto, el número asignado como nodo de referencia en cada circuito de secuencia no puede ser cero. En este caso se asignaron los números 100, 200 y 300 como nodos de referencia en los circuitos de secuencia positiva, negativa y cero, respectivamente.




Figura 5. Circuito de secuencia positiva modificado para PSPICE




Figura 6. Circuito de secuencia negativa modificado para PSPICE




Figura 7. Circuito de secuencia cero modificado para PSPICE

García y Rivera/Simulación fallas componentes simétricos

La definición de los circuitos de secuencia como subcircuitos permite que los números asignados a los nodos se puedan repetir en otro subcircuito. De la misma forma que el programa PSPICE distingue los nombres de las fuentes de voltaje que se repiten en los tres circuitos de secuencia durante el análisis, lo hace con los nodos. Por ejemplo, X1.3, X2.3 y X3.3, donde X1.3 significa nodo número tres definido en el subcircuito que llama la instrucción X1, X2.3 es el nodo número tres definido en el subcircuito que llama la instrucción X2 y X3.3 es el nodo número tres definido en el subcircuito que llama la instrucción X3.

Eliminar todo elemento que no tenga un paso de corriente directa al nodo de referencia

Ningún elemento del circuito debe crear un circuito abierto. Esta restricción aplica al circuito de secuencia cero, donde el tipo de conexión de los embobinados primarios y secundarios de los bancos de transformadores ocasiona que el flujo de corriente de secuencia cero entre el embobinado primario y el secundario cese. Por esta razón, los bancos de transformadores se representan como un circuito abierto en sus terminales.

Crear el archivo que defina los circuitos de secuencia positiva, negativa y cero como subcircuitos

El archivo que define los circuitos de secuencia forma parte de la biblioteca de PSPICE y se le asigna el nombre de SEC.LIB. Durante la explicación de las partes que componen este archivo se hace referencia a este listado que aparece en el apéndice 1.

Cada subcircuito se define en PSPICE con la instrucción .SUBCKT

El formato general de la instrucción .SUBCKT es el siguiente:

.SUBCKT SUBNAM N1 N2

donde

- .SUBCKT - indica definición de un subcircuito
- SUBNAM - nombre asignado a los circuitos de secuencia (SECPOS , SECNEG y SECCERO)
- N1 - Nodo externo. Representa el lugar donde ocurre la falla en el circuito de secuencia. Si se desea cambiar el lugar donde ocurre la falla en la simulación es necesario cambiar el numero de N1. En este caso que se desea simular una falla en la barra 6 del sistema de potencia de la figura 1, el valor de N1 debe ser cuatro en la definición de cada uno de los subcircuitos (figs. 5,6 y 7).
- N2 - Nodo externo. Indica el nodo de referencia en el circuito de secuencia.

Descripción topológica de los subcircuitos (circuitos de secuencia)

En general, en un análisis de corto circuito, el componente resistivo de las líneas de transmisión, transformadores, generadores o motores se puede descartar por ser de una magnitud mucho menor que el componente inductivo ($X_L \ll R$). De esta manera se reduce la cantidad de variables usadas en el proceso matemático. PSPICE usa muchas restricciones en las características topológicas de un circuito. El circuito no tendrá lazos ("loops") de fuentes de voltaje y de inductores sin un elemento resistivo. Por lo tanto, el usuario debe evitar realizar simulaciones que omitan el componente resistivo. Una alternativa es añadir una resistencia de valor considerablemente bajo en serie con el elemento inductivo.

Es importante señalar que la reactancia inductiva dada en por unidad (p.u.) debe convertirse a unidades de inductancia (Henrios) mediante la formula $L = X_L / 2\pi f$ para PSPICE, donde L es la inductancia en Henrios, X_L es la reactancia inductiva en Ohmios y F es la frecuencia del sistema Hertz.

García y Rivera/Simulación fallas componentes simétricos

Si definimos $2\pi f = 376.991118431$ como un parámetro para calcular la inductancia, este parámetro hay que definirlo en la instrucción que define el subcircuito con el siguiente formato:

```
.SUBCKT SUBNAM N1 N2 PARAMS: A = 376.991118431
```

Una vez definido el parámetro, éste se puede usar en la definición de un elemento inductivo.

Final de la definición del subcircuito.

La descripción topológica de cada subcircuito debe terminar con la instrucción .ENDS, cuyo formato es:

```
.ENDS SUBNAM
```

donde ".ENDS" indica el final de la definición del subcircuito y SUBNAM es el nombre del subcircuito que se define. Si se omite el nombre del subcircuito, entonces se considera el final de todos los subcircuitos definidos anteriormente.

Una vez se definen los circuitos de secuencia positiva, negativa y cero como subcircuitos, este archivo pasa a formar parte de la biblioteca de PSPICE con la extensión "LIB". La extensión "LIB" le indica al programa PSPICE que este archivo es parte de su biblioteca. El nombre que le asignamos a este archivo es "SEC.LIB".

Crear el archivo fuente para llamar e interconectar los subcircuitos de la biblioteca de PSPICE y formar el circuito global que simula la falla

Hay que crear un archivo fuente por cada tipo de falla que se desee simular. El apéndice 2 presenta el listado de un archivo fuente para simular una falla asimétrica doble de línea a tierra.

Los archivos fuentes se componen principalmente de seis partes. Como guía se usa el diagrama de la figura 8 para crear cada archivo. Este tipo de diagrama representa el equivalente de Thevenin mirando en el punto donde ocurre la falla.




Figura 8. Equivalente de Thevenin para la falla doble de línea tierra

García y Rivera/Simulación fallas componentes simétricos

$V_{prefalla}$: es el voltaje antes de que ocurra la falla

VD1, VD2 y VD3

VD1, VD2 y VD3 son fuentes de voltaje DC de cero voltios utilizadas como amperímetro para medir los componentes simétricos de la corriente de falla. VD1 se usa para para medir el componente positivo de las corrientes desbalanceadas de falla y VD2 y VD3 se usan para medir los componentes negativos y cero, respectivamente.

La polaridad de las fuentes de voltaje la determina el orden en que se especifican los nodos en la definición de la fuente. Esto significa que el primer nodo es positivo con respecto al segundo. En el caso de VD1, VD2 y VD3, PSPICE presume que las corrientes fluyen entrando por el nodo positivo. Por lo tanto, estas fuentes deben especificarse en la dirección que aporten corriente al punto de falla según lo muestra la figura 8.

Llamar e interconectar los subcircuitos para formar el circuito global que simula la falla

La instrucción que llama e interconecta los subcircuitos debe comenzar con la letra X y su formato general es:

$X_n M1 M2 SUBNAM$

donde X_n indica que se llama un subcircuito, $X1$ llama el circuito de secuencia positiva, $X2$ llama el circuito de secuencia negativa y $X3$ llama el circuito de secuencia cero. $M1$ y $M2$ son los nodos de conexión externos del subcircuito. Estos nodos pertenecen al circuito global donde se conectan los subcircuitos. $SUBNAM$ es el nombre del subcircuito que se llama.

La figura 8 muestra el diagrama de interconexión del subcircuito para la simulación de la falla. Los nodos enumerados dentro de las líneas entrecortadas representan los nodos externos definidos en cada subcircuito por la instrucción `.SUBCKT` (contenida en el archivo `sec.lib`). Los nodos

enumerados fuera de las líneas entrecortadas son los nodos del circuito global definidos en la instrucción que llama los subcircuitos X_n (contenida en el archivo fuente). Podemos observar que la cantidad de nodos definidos en la instrucción X_n es igual a la cantidad de nodos definidos por la instrucción .SUBCKT y que los nodos se conectan en el orden en que se definen. Esto significa que el primer nodo definido en la instrucción que define el subcircuito (.SUBCKT) se conecta al primer nodo definido en la instrucción que llama (X_n) los subcircuitos y el segundo nodo con el segundo.

Es importante señalar que el circuito global formado por los subcircuitos (circuitos de secuencia positiva, negativa y cero) debe tener el nodo de referencia cero para que el programa PSPICE pueda hacer el análisis.

Indicarle al programa que los subcircuitos que se llaman forman parte de la biblioteca de PSPICE y están almacenados en uno de sus archivos

Por ejemplo, en la instrucción .LIB B:SEC.LIB, donde "LIB" le indica al programa que los subcircuitos que se llaman forman parte de la biblioteca de PSPICE; "B:" indica la unidad de disco (A, B, C) y SEC.LIB es el nombre del archivo que contiene los subcircuitos.

Tipo de análisis

Una vez se llaman e interconectan los subcircuitos que forman el circuito global que simula la falla, es necesario seleccionar en PSPICE el tipo de análisis que se desea hacer. En este caso se hizo un análisis de corriente alterna (AC) a una frecuencia de 60Hz. El formato de la instrucción para este tipo de análisis es el siguiente:

.AC LIN 1 60 60

Información a imprimir de la simulación.

La información a imprimirse de la simulación son los componentes simétricos de la corriente que fluye en la falla y en cualquier otro punto del

García y Rivera/Simulación fallas componentes simétricos

sistema de potencia. Si se desea imprimir la magnitud y el ángulo de fase de los componentes simétricos de la corriente que fluye en la falla, el formato es el siguiente:

```
.PRINT AC IM(VD1) IP(VD1) IM(VD2) IP(VD2)  
+IM(VD3) IP(VD3)
```

donde VD1, VD2 Y VD3 son las fuentes de voltaje "CC" del circuito global que mide el componente positivo, negativo y cero de la corriente desbalanceada en la falla.

De igual forma se puede imprimir la magnitud y el ángulo de los componentes simétricos de la corriente que fluye en otros puntos del sistema de potencia. Por ejemplo, si desea medir la corriente de los componentes positivos, negativos y cero que fluyen del nodo 1 al 6 del sistema de potencia de la figura 1, el formato es:

```
.PRINT AC IM(X1.VD1) IP(X1.VD1) IM(X2.VD1) IP(X2.VD1)  
+IM(X3.VD1) IP(X3.VD1)
```

La tabla 2 presenta las corrientes de cada secuencia medidas en la simulación y el formato en PSPICE para imprimir cada una de las corrientes mediante la instrucción ".PRINT".

Resultados de la simulación

Luego que PSPICE analiza el circuito global que simula la falla, éste crea un archivo de salida cuyo nombre es el mismo que el archivo fuente pero su extensión es .OUT. Este archivo resume toda la información del archivo fuente, el tipo de análisis y la información que mediante la instrucción ".PRINT" se desea imprimir de la simulación. Los valores que se desean imprimir de las simulaciones son la magnitud y el ángulo de fase de los componentes simétricos de las corrientes en el sistema.

PSPICE solamente permite imprimir los componentes simétricos de la corriente. No dispone de instrucciones para calcular las corrientes desbalanceadas de las tres fases del sistema. Una vez se completa la simulación, se usa la ecuación (1) para calcular las corrientes desbalanceadas en las tres fases del sistema (en por unidad).

Tabla 2. Corrientes a imprimir y su formato en PSPICE

Corriente medida	Formato en PSPICE
I_f^0	IM(VD3) IP(VD3)
I_f^1	IM(VD1) IP(VD1)
I_f^2	IM(VD2) IP(VD2)
I_{1-6}^0	IM(X3.VD1) IP(X3.VD1)
I_{1-6}^1	IM(X1.VD1) IP(X1.VD1)
I_{1-6}^2	IM(X2.VD1) IP(X2.VD1)
I_{4-6}^0	IM(X3.VD2) IP(X3.VD2)
I_{4-6}^1	IM(X1.VD2) IP(X1.VD2)
I_{4-6}^2	IM(X2.VD2) IP(X2.VD2)
I_{5-6}^0	IM(X3.VD3) IP(X3.VD3)
I_{5-6}^1	IM(X1.VD3) IP(X1.VD3)
I_{5-6}^2	IM(X2.VD3) IP(X2.VD3)

García y Rivera/Simulación fallas componentes simétricos

$$\begin{bmatrix} I_a \\ I_b \\ I_c \end{bmatrix} = \begin{bmatrix} 1 & 1 & 1 \\ 1 & a^2 & a \\ 1 & a & a^2 \end{bmatrix} \begin{bmatrix} I_a^0 \\ I_a^1 \\ I_a^2 \end{bmatrix} \quad (1)$$

donde a es un operador cuyo valor es:

$$a = 1 \angle 120^\circ \quad \text{y} \quad a^2 = 1 \angle 240^\circ \quad (2)$$

Para calcular las corrientes desbalanceadas en amperios se multiplica esta corriente por la corriente base.

Apéndice 1

*DEFINICION DE LOS CIRCUITOS DE SECUENCIA POSITIVA,
NEGATIVA Y CERO COMO SUBCIRCUITOS*

*****SECUENCIA POSITIVA*****

.SUBCKT SECPOS 4 100 PARAMS: A=376.991118431

LXG1 1 2 {66.6666666666M/A}

RG1 1 100 1.33333333333333M

RT1 2 3 1.42857142857M

LXT1 3 17 {25.7142857143M/A}

VD1 4 17 DC 0

VD2 4 18 DC 0

VD3 4 19 DC 0

RL56 19 5 26.2549884478M

LXL56 5 6 {525.099768956M/A}

LXT2 6 7 {80.0M/A}

RT2 7 8 5.33333333333334M

LXG2 8 9 {150.0M/A}

RG2 9 100 5.0M

RL46 18 10 10.5019953791M

LXL46 10 11 {420.079815165M/A}

LXL45 11 12 {472.58979206M/A}

RL45 12 6 21.0039907582M

LXT3 11 13 {40.0M/A}

RT3 13 14 2.0000M

LXG3 14 15 {90.0M/A}

RG3 15 100 3.00M

.ENDS SECPOS

García y Rivera/Simulación fallas componentes simétricos

*****SECUENCIA NEGATIVA*****

```
.SUBCKT SECNEG 4 200 PARAMS: A=376.991118431
LXG1 1 2 {83.333333332M/A}
RG1 1 200 1.6666666666666M
RT1 2 3 1.42857142857M
LXT1 3 17 {25.7142857143M/A}
VD1 17 4 DC 0
VD2 18 4 DC 0
VD3 19 4 DC 0
RL56 19 5 26.2549884478M
LXL56 5 6 {525.099768956M/A}
LXT2 6 7 {80.0M/A}
RT2 7 8 5.33333333334M
LXG2 8 9 {200.0M/A}
RG2 9 200 6.00M
RL46 18 10 10.5019953791M
LXL46 10 11 {420.079815165M/A}
LXL45 11 12 {472.589792060M/A}
RL45 12 6 21.0039907582M
LXT3 11 13 {40.0M/A}
RT3 14 13 2.00M
LXG3 14 15 {90.0M/A}
RG3 15 200 3.000M
.ENDS SECNEG
```

*****SECUENCIA CERO*****

```
.SUBCKT SECCERO 4 300 PARAMS: A=376.991118431
RT1 3 300 1.42857142857M
LXT1 3 17 {25.7142857143M/A}
VD1 17 4 DC 0
VD2 18 4 DC 0
VD3 19 4 DC 0
RL56 19 5 26.2549884478M
```

LXL56 5 6 {525.099768956M/A}
LXT2 6 7 {80.0M/A}
RT2 7 300 5.33333333333334M
RL46 18 10 10.5019953791M
LXL46 10 11 {525.099768956M/A}
LXL45 11 12 {525.099768956M/A}
RL45 12 6 21.0039907582M
LXT3 11 14 {40.0M/A}
RT3 14 300 2.0M
.ENDS SECCERO
.END

García y Rivera/Simulación fallas componentes simétricos

Apéndice 2:

FALLA DOBLE LINEA A TIERRA

*****VOLTAJE PREFALLA*****

VPREFALLA 1 0 AC 1 0

VD1 1 2 DC 0

VD2 4 3 DC 0

VD3 5 3 DC 0

*****LLAMAR SUBCIRCUITOS*****

X1 2 3 SECPOS

X2 4 0 SECNEG

X3 5 0 SECCERO

*****SUBCIRCUITOS DE LA LIBRERIA DE PSPICE*****

***INDICADO POR LA INSTRUCCION .LIB, EN LA UNIDAD DE

DISCO B:***

.LIB B:SEC.LIB

*****ANALISIS AC FRECUENCIA 60 Hz*****

.AC LIN 1 60 60

*****INFORMACION QUE SE DESEA IMPRIMIR*****

*****CORRIENTES DE SECUENCIA POSITIVA*****

.PRINT AC IM(VD1) IP(VD1) IM(VD2) IP(VD2) IM(VD3) IP(VD3)

.PRINT AC IM(X1.VD1) IP(X1.VD1) IM(X1.VD2) IP(X1.VD2)

+IM(X1.VD3) IP(X1.VD3)

*****CORRIENTES DE SECUENCIA NEGATIVA*****

```
.PRINT AC IM(X2.VD1) IP(X2.VD1) IM(X2.VD2) IP(X2.VD2)
+IM(X2.VD3) IP(X2.VD3)
```

*****CORRIENTES DE SECUENCIA CERO*****

```
.PRINT AC IM(X3.VD1) IP(X3.VD1) IM(X3.VD2) IP(X3.VD2)
+IM(X3.VD3) IP(X3.VD3)
```

*****FINAL DEL ARCHIVO*****

```
.OPTION NOBIAS
.END
```